********* TABLE 2 *********


clear all
set more off 
capture log close
 
* main path
global path "~/Dropbox/Paraguay/1_data/7_replication_dataverse"
capture cd "$path"	
*tables = analysis tables
global tables "$path/tables"


********** PANEL A: LEGITIMATE GRANTS **********

* Load and prepare data
use "$path/data/indert.dta", clear

xtset id year
generate autocracy = (year<1989)

eststo clear

* Loop over dependent variables for Panel A
foreach y in grants_ha_log grants_e {
    local i = 1
    foreach fe in "year id" "year id i.depa#i.year" "year id i.district_code#i.year" "year id i.district_code#i.year i.year#i.lastname_d" {
        reghdfe `y' i.autocracy##c.zeigen [aweight=p_weights] if sample_weight == 1, absorb(`fe') cluster(id)
        
        * Store fixed effects indicators
        estadd local DT = cond("`fe'"=="year id i.depa#i.year", "Yes", "No")
        estadd local MY = cond("`fe'"=="year id i.district_code#i.year" | "`fe'"=="year id i.district_code#i.year i.year#i.lastname_d", "Yes", "No")
        estadd local FM = cond("`fe'"=="year id i.district_code#i.year i.year#i.lastname_d", "Yes", "No")
        
        * Store mean of dependent variable
        summ `y' if e(sample)==1 
        estadd scalar meanDep = r(mean)
        
        local col = `i'
        if "`y'" == "grants_e" local col = `i' + 4
        eststo legit`col'
        local ++i
    }
}

********** PANEL B: ILL-GOTTEN LANDS **********
* Load and prepare data
use "$path/data/main.dta", clear

xtset id year
generate autocracy = (year<1989)

* Drop departments not in placebo sample
foreach dept in "caaguazu / yhu" "caazapa" "cordillera / caraguatay" "guaira" ///
    "itapua / encarnacion" "misiones / san ignacio" "neembucu / pilar" ///
    "paraguari / quiindy" "seccion occidental / chaco" {
    drop if departamento == "`dept'"
}

* Loop over dependent variables for Panel B
foreach y in tmh_ha_log tmh_e {
    local i = 1
    foreach fe in "year id" "year id i.depa#i.year" "year id i.district_code#i.year" "year id i.district_code#i.year i.year#i.lastname_d" {
        reghdfe `y' i.autocracy##c.zeigen [pweight=p_weights] if sample_weight == 1, absorb(`fe') cluster(id)
        
        * Store fixed effects indicators
        estadd local DT = cond("`fe'"=="year id i.depa#i.year", "Yes", "No")
        estadd local MY = cond("`fe'"=="year id i.district_code#i.year" | "`fe'"=="year id i.district_code#i.year i.year#i.lastname_d", "Yes", "No")
        estadd local FM = cond("`fe'"=="year id i.district_code#i.year i.year#i.lastname_d", "Yes", "No")
        
        * Store mean of dependent variable
        summ `y' if e(sample)==1 
        estadd scalar meanDep = r(mean)
        
        local col = `i'
        if "`y'" == "tmh_e" local col = `i' + 4
        eststo illgot`col'
        local ++i
    }
}

* Create table 
file open table2 using "$tables/table2.tex", write replace
file write table2 "\begin{tabular}{lcccclcccc} \hline" _n
file write table2 "& \multicolumn{4}{c}{Hectares (log)}  &  & \multicolumn{4}{c}{Land Grant (binary)} \\" _n  
file write table2 "[3pt] \cline{2-5} \cline{7-10}\\" _n
file write table2 " & (1)        & (2)        & (3)        & (4)       &  & (5)        & (6)        & (7)        & (8)      \\" _n  
file write table2 "\hline" _n
file write table2 "\multicolumn{10}{l}{\textit{(a) Legitimate (to Peasants)}} \\" _n
file close table2

* Panel A results with R2 and outcome mean
esttab legit* using "$tables/table2.tex", append fragment nonumber label style(tex) ///
    cells(b(star fmt(%12.4fc)) se(par fmt(%12.4fc))) ///
    keep(1.autocracy#c.zeigen) ///
    stats(r2 meanDep, fmt(%12.4fc %12.4fc) label("\$R^2\$" "Outcome mean")) ///
    varlabels(1.autocracy#c.zeigen "Autocracy \$\times\$ Eigenvector") ///
    extracols(5) collabels(none) mlabels(none) eqlabels(none) starlevels(* 0.1 ** 0.05 *** 0.01) ///
    noobs nonotes nogaps noline prehead("") posthead("") prefoot("") postfoot("")

* Add Panel B header
file open table2 using "$tables/table2.tex", write append
file write table2 "\hline" _n
file write table2 "\multicolumn{10}{l}{\textit{(b) Ill-Gotten (to Elites)}} \\" _n
file close table2

* Panel B results with R2 and outcome mean
esttab illgot* using "$tables/table2.tex", append fragment nonumber label style(tex) ///
    cells(b(star fmt(%12.4fc)) se(par fmt(%12.4fc))) ///
    keep(1.autocracy#c.zeigen) ///
    stats(r2 meanDep, fmt(%12.4fc %12.4fc) label("\$R^2\$" "Outcome mean")) ///
    varlabels(1.autocracy#c.zeigen "Autocracy \$\times\$ Eigenvector") ///
    extracols(5) collabels(none) mlabels(none) eqlabels(none) starlevels(* 0.1 ** 0.05 *** 0.01) ///
    noobs nonotes nogaps noline prehead("") posthead("") prefoot("") postfoot("")

* Add bottom statistics in a separate esttab command
esttab illgot* using "$tables/table2.tex", append fragment nonumber label style(tex) ///
    cells(none) ///
    stats(N DT MY FM, fmt(%12.0fc 0 0 0) ///
    labels("Observations" "Department-Year FE" "Municipality-Year FE" "Family-Year FE")) ///
    extracols(5) collabels(none) mlabels(none) eqlabels(none) ///
    noobs nonotes nogaps noline prehead("") posthead("") prefoot("\hline") postfoot("\hline \end{tabular}")
	
